

Lab 13, interrupts, part 2

By ADMIN | Published: APRIL 14, 2013

Time to write a code to process interrupts.

All interrupts that happens in system use the vector of interrupts (8 of them) which is located at **0xffff0000** (high memory case). This vector contains addresses that CPU should pass execution to.

1.	+00	Reset
2.	+04	Undefined
3.	+08	SWI
4.	+0C	Prefetch abort
5.	+10	Data abort
6.	+14	Reserved
7.	+18	IRQ
8.	+1C	FIQ

In *main()* we have to call function called *trapinit()* to do all initializations of interrupts handling.

Example: when interrupt happens – say IRQ (number 7), then cpu will pass execution to **0xffff0000+18**. Our task is to initialize that place with codes that will pass execution to specific calls in kernel.

It is done in next way: in assembler file **intr.s** we create two functions:

```
01 | TEXT vectors(SB), $-4
02 |     MOVW    0x18(PC), PC    /* reset */
03 |     MOVW    0x18(PC), PC    /* undefined */
04 |     MOVW    0x18(PC), PC    /* SWI */
05 |     MOVW    0x18(PC), PC    /* prefetch abort */
06 |     MOVW    0x18(PC), PC    /* data abort */
07 |     MOVW    0x18(PC), PC    /* reserved */
08 |     MOVW    0x18(PC), PC    /* IRQ */
09 |     MOVW    0x18(PC), PC    /* FIQ */
10 |
11 | TEXT vtable(SB), $-4
12 |     WORD    $_vsvc(SB)      /* reset, in svc mode already */
13 |     WORD    $_vund(SB)      /* undefined, switch to svc mode */
14 |     WORD    $_vsvc(SB)      /* swi, in svc mode already */
15 |     WORD    $_vpab(SB)      /* prefetch abort, switch to svc mode */
16 |     WORD    $_vdab(SB)      /* data abort, switch to svc mode */
17 |     WORD    $_vsvc(SB)      /* reserved */
18 |     WORD    $_virq(SB)      /* IRQ, switch to svc mode */
19 |     WORD    $_vfirq(SB)     /* FIQ, switch to svc mode */
20 |
21 | TEXT _vund(SB), $-4
22 |     ...
```

- Categories
- [Blog](#)
 - [Boost](#)
 - [C++](#)
 - [Cryptography](#)
 - [Embedding](#)
 - [Hybrids](#)
 - [Inferno OS](#)
 - [MacAppStore](#)
 - [Misc](#)
 - [Models](#)
 - [Projects](#)
 - [PyQt](#)
 - [PySide](#)
 - [Qt](#)
 - [QtSpeech](#)
 - [Raspberry Pi](#)
 - [Research](#)
 - [Ru](#)
 - [TogMeg](#)
 - [Trac](#)
 - [TTS](#)
 - [Tutorial](#)
 - [Undo](#)
 - [Web](#)

Then in *trapinit()* we copy bytes of these functions to **0xffff0000**:

```
01 enum { Nvec = 8 }; /* # of vectors */
02 typedef struct Vpage0 {
03     void      (*vectors[Nvec])(void);
04     u32int    vtable[Nvec];
05 } Vpage0;
06
07 void trapinit(void) {
08     Vpage0 *vpage0;
09     /* set up the exception vectors */
10     vpage0 = (Vpage0*)HVECTORS;
11     memmove(vpage0->vectors, vectors, sizeof(vpage0->vectors));
12     memmove(vpage0->vtable,  vtable,  sizeof(vpage0->vtable));
13     ...
```

You may see that when execution passed to **0xffff0000+18** (HVECTORS+IRQ), then cpu does *MOVW 0x18(PC), PC*, which means to jump to address which is located in memory just +0x18 above – where the vtable with addresses of our kernel routines *_virq()*

Then our 8 assembler routines to do initial logic of handling interrupts”

```
01 TEXT _vund(SB), $-4
02     MOVMD.B [R0-R3], (SP)
03     MOVW    $PsrMund, R0
04     B       _vswitch
05
06 TEXT _vsvc(SB), $-4
07     MOVW.W  R14, -4(SP)
08     MOVW    CPSR, R14
09     MOVW.W  R14, -4(SP)
10     BIC     $PsrMask, R14
11     ORR     $(PsrDirq|PsrDfiq|PsrMsvc), R14
12     MOVW    R14, CPSR
13     MOVW    $PsrMsvc, R14
14     MOVW.W  R14, -4(SP)
15
16     B       _vsaveu
17
18 TEXT _vpab(SB), $-4
19     MOVMD.B [R0-R3], (R13)
20     MOVW    $PsrMabt, R0
21     B       _vswitch
22
23 TEXT _vdab(SB), $-4
24     MOVMD.B [R0-R3], (R13)
25     MOVW    $(PsrMabt+1), R0
26     B       _vswitch
27
28 TEXT _vfiq(SB), $-4          /* FIQ */
29     MOVMD.B [R0-R3], (R13)
30     MOVW    $PsrMfiq, R0
31     B       _vswitch
32
33 TEXT _virq(SB), $-4          /* IRQ */
34     MOVMD.B [R0-R3], (R13)
35     MOVW    $PsrMirq, R0
36
37 _vswitch: /* switch to svc mode */
```

You see that they are using 4 words of stack *[R0-R3]*. That I will show later in *trapinit()*, but idea that we can set different stack addresses for each interrupt types – that are those arrays we added to *Mach*: *ulong fiqstack[4]; ulong irqstack[4]; ulong abtstack[4]; ulong undstack[4];*

Later it switches to **svc** mode and cpu gets stack with address which is set for svc mode.

```
01 _vswitch:
02     MOVW    SPSR, R1          /* switch to svc mode */
03     MOVW    R14, R2          /* state of cpu, cpsr */
04     MOVW    SP, R3           /* return code */
05
06     MOVW    CPSR, R14        /* stack */
07     BIC     $PsrMask, R14
08     ORR     $(PsrDirq|PsrDfiq|PsrMsvc), R14
09     MOVW    R14, CPSR        /* switch! */
10
11     MOVW    R0, R0           /* gratuitous noop */
12
```

```

13      MOVW.DB.W      [R0-R2], (SP)      /* set ureg->{type, psr, pc}; */
14                                          /* SP points to ureg->type */
15      MOVW           R3, -12(SP)
16      MOVW.IA       (R3), [R0-R3]      /* restore [R0-R3] from previous mode */
17
18  _vsaveu:
19      MOVW.W        R11, -4(SP)        /* save link */
20
21      SUB           $8, SP
22      MOVW.DB.W     [R0-R12], (SP)
23
24      MOVW          $setR12(SB), R12/* Make sure we've got the kernel's SB */
25      MOVW          SP, R0             /* first arg is pointer to ureg */
26      SUB           $(4*2), SP        /* space for argument+link (for debugger) */
27      MOVW          $0xdeaddead, R11/* marker */
28      BL            trap(SB)
29
30  _vrfe:
31      MOVW          CPSR, R0           /* Restore Regs */
32                                          /* splhi on return */
33      ORR           $(PsrDirq|PsrDfiq), R0, R1
34      MOVW          R1, CPSR
35      ADD           $(8+4*15), SP      /* [r0-R14]+argument+link */
36      MOVW          (SP), R14          /* restore link */
37      MOVW          8(SP), R0
38      MOVW          R0, SPSR
39      MOVW.DB.S     (SP), [R0-R14]    /* restore user registers */
40      MOVW          R0, R0             /* gratuitous nop */
41      ADD           $12, SP            /* skip saved link+type+SPSR*/
42      RFE
43                                          /* MOVW.IA.S.W (SP), [PC] */

```

In codes above we can use *R0-R3* as we saved then in Mach object. With 4 registers we can perform disabling interrupts (svc), next CPU state saving into Ureg object, adjusting *SB* and finally pass the control to *trap()* C-routine of kernel to process the exception.

On return from *trap()* – **_vrfe**, it does opposite, restore registres from Ureg object, enables interrupts and pass control back to the point where execution was interrupted.

FILES:

- [intr.s](#)
- [trap.c](#)
- [mkfile](#)

This entry was posted in *Blog, Inferno OS, Raspberry Pi, Research*. Bookmark the *permalink*. *Post a comment* or leave a *trackback*: *Trackback URL*.

